

目 录

第一章 LotusScript Notes类参考	1
Button 类	1
Field 类	2
Navigator 类	4
NotesACL 类	4
NotesACLEntry 类	16
NotesAgent 类	31
NotesDatabase 类	43
NotesDateRange 类	112
NotesDateTime 类	115
NotesDbDirectory 类	135
NotesDocument 类	138
NotesDocumentCollection 类	202
NotesEmbeddedObject 类	216
NotesForm 类	228
NotesInternational 类	238
NotesItem 类	249
NotesLog 类	272
NotesName 类	292
NotesNewsletter 类	300
NotesRichTextItem 类	308
NotesSession 类	323
NotesTimer 类	353
NotesUIDatabase 类	358
NotesUIDocument 类	364
NotesUIView 类	412
NotesUIWorkspace 类	422
NotesView 类	437

NotesViewColumn 类	469
ODBCConnection 类	479
ODBCQuery 类	492
ODBCResultSet 类	499
第二章 带有ECL安全性的属性和成员函数	538

第一章 LotusScript Notes类参考

本章列举了 Notes 类库中每个类、属性和成员函数的信息。每个类的后面都按字符顺序列出了它的属性和成员函数名。

Button类

代表表单或文档中的操作、按钮或热点。

事件

Click

ObjectExecute

访问

要访问操作、按钮、或热点，请编写响应它们的事件的 script。一个 Button 对象没有属性或成员函数。

用法

按钮、域和导航器事件提供了从 Notes UI 到 scripts 的入口。

Click 事件

当用户选择操作、按钮或热点时发生。

定义

Button 类

语法

Click(Source as Button)

参数

Source

按钮。只读。操作、按钮或热点。

用法

当您创建响应本事件的 script 时，Notes 为您创建该事件的语法（包括参数）。无须输入。

ObjectExecute 事件

当操作、按钮或热点被 OLE2 服务器（FX/Notesflow 是激活）激活时发生。

定义

Button 类

语法

ObjectExecute(Source as Button, Data)

参数

Source

按钮。只读。

用法

当您创建响应本事件的 script 时，Notes 为您创建该事件的语法（包括参数）。无须输入。

Field 类

代表表单中的一个域。

事件

Entering

Exiting

访问

要访问域，请编写响应其事件的 script。一个 Field 对象没有属性或成员函数。

用法

按钮、域和导航器事件提供了从 Notes UI 到 scripts 的进入点。

Entering 事件

当用户在编辑模式下输入一个域时发生。

定义

Field 类

语法

Entering(Source as Field)

参数

Source

域。只读。

用法

当您创建响应本事件的 script 时，Notes 为您创建该事件的语法（包括参数）。无须输入。

Exiting 事件

当用户在编辑模式下退出一个域时发生。

定义

Field 类

语法

Exiting(Source as Field)

参数

Source

域。只读。

用法

当您创建响应本事件的 script 时，Notes 为您创建该事件的语法（包括参数）。无须输入。

Navigator 类

代表一个导航器按钮、热点或其他对象。

事件

Click

访问

要访问一个导航器对象，请编写一个响应其事件的 script。导航器对象没有属性或成员函数。

用法

按钮、域和导航器事件提供了从 Notes UI 到 scripts 的进入点。

Click 事件

用户选择一个导航器对象时发生。

定义

Navigator 类

语法

Click(Source as Navigator)

参数

Source

导航器。只读。导航器对象。

用法

当您创建响应本事件的 script 时，Notes 为您创建该事件的语法（包括参数）。无须输入。

NotesACL 类

代表数据库的存取控制表 (ACL)。

包容

包容在: NotesDatabase

包容: NotesACLEntry

属性

Parent

Roles

UniformAccess

成员函数

AddRole

CreateACLEntry

DeleteRole

GetEntry

GetFirstEntry

GetNextEntry

RenameRole

Save

每个 NotesDatabase 都包含一个 NotesACL 对象来代表数据库的存取控制表。要取得它, 请使用 NotesDatabase 的 ACL 属性。

用法

为了方便使用, NotesDatabase 类中有三个成员函数可以用来访问和修改 ACL, 而不必声明 NotesACL 对象。您需要知道的只是个人、服务器和群组的名字。

- 查看个人、服务器和群组的存取权限的级别, 请使用 QueryAccess。
- 要授予存取权限, 请使用 GrantAccess。
- 要删除存取权限, 请使用 RevokeAccess。

样例: NotesACL 类

1. 该 script 获取在服务器 Maputo 上的数据库的 ACL, 且使用 GetFirstEntry 成员函数找出 ACL 中的第一个条目。

```
Dim db As New NotesDatabase( "Maputo", "notefile\none.nsf" )
Dim acl As NotesACL
Dim entry As NotesACLEntry
Set acl = db.ACL
Set entry = acl.GetFirstEntry
```

2. 该 script 获取当前数据库的 ACL, 且使用 GetEntry 找出 ACL 中的 Dyetra Jones。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Dim entry As NotesACLEntry
Set db = session.CurrentDatabase
Set acl = db.ACL
Set entry = acl.GetEntry( "Dyetra Jones" )
```

Parent 属性

只读。拥有该 ACL 的数据库。

定义

NotesACL

数据类型

NotesDatabase

语法

取得: **Set notesDatabase = notesACL.Parent**

样例: Parent 属性

本 script 打印 NotesDatabase 对象 (db) 的标题 (从该数据库中提取了 ACL), 以及由 Parent 属性 (parentDb) 返回的 NotesDatabase 对象的标题。标题是相同的, 是因为 db 和 parentDb 代表同一个数据库 SOMEDOC.SNSF。

```
Dim db As New NotesDatabase( "", "somedoc.nsf" )
Dim parentDb As NotesDatabase
Dim acl As NotesACL
Set acl = db.ACL
Set parentDb = acl.Parent
MessageBox( db.Title )
MessageBox( parentDb.Title )
```

Roles 属性

只读。定义在存取控制表中的所有角色。

定义

NotesACL

数据类型

字符串数组

语法

取得: `stringArray = notesACL.Roles`

用法

数组中的每一个元素都是一个角色名称。每一个角色名称都被圆括号或方括号括住: 例如, [Supervisor]。

样例: Roles 属性

1. 本 script 显示当前数据库中的每一个角色名字。例如: 可能显示 [HR Contact], [Supervisor] 和 [Project Leader]。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Set db = session.CurrentDatabase
Set acl = db.ACL
Forall r In acl.Roles
    Messagebox( r )
End Forall
```

2. 本 script 让用户在对话框中输入一个角色名字。然后该 script 显示当前文档的 People 域中具有该角色的每个条目名称。该 script 在用户输入的角色名上添加方括号, 并使用 Roles 属性在当前数据库中确认该名字是否存在。如果没有, 将显示一个错误信息; 如果有, 则 script 继续执行。

```
Dim workspace As New NotesUIWorkspace
Dim uidoc As NotesUIDocument
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Dim entry As NotesACLEntry
Dim roleName As String
Dim foundRole As Variant
Set uidoc = workspace.CurrentDocument
Set db = session.CurrentDatabase
Set acl = db.ACL
roleName = Inputbox$( "Enter the name of the role" )
' add brackets to role
roleName = "[" & roleName & "]"
foundRole = False
' check to see if the role exists in the database
Forall r In acl.Roles
    If ( r = roleName ) Then
        foundRole = True
    Exit Forall
End If
```

```
End Forall
If ( foundRole = False ) Then
  Messagebox _
    ( "Sorry, " & roleName & " is not a role" )
  ' if the role exists, check each acl entry to see if role
  ' is enabled for entry
  ' if so, add entry name to the People field
  ' on the current document,
  ' followed by a semicolon, the multi-value separator
Else
  Set entry = acl.GetFirstEntry
  While Not ( entry Is Nothing )
    If ( entry.IsRoleEnabled( roleName ) = True ) Then
      Call uidoc.FieldAppendText _
        ( "People", entry.Name & ";" )
    End If
    Set entry = acl.GetNextEntry( entry )
  Wend
End If
' refresh current document so People field displays nicely
Call uidoc.Refresh
```

UniformAccess 属性

可读写。指出是否设置了统一的存取权限。

定义

NotesACL 类

数据类型

布尔值

语法

取得: *flag* = *notesACL.UniformAccess*

设置: *notesACL.UniformAccess* = *flag*

合法值

True	统一的存取权限已设置
False	统一的存取权限没有设置

用法

在 Notes UI 中, 这个属性在数据库存取控制表的高级区段中: “此数据库的所有复本使用相同的存取控制表”。

样例: UniformAccess 属性

以下样例从用户的角度说明 UniformAccess 的属性。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Set db = session.CurrentDatabase
Set acl = db.ACL
If acl.UniformAccess Then
    If MessageBox("Do you want to toggle?", _
        MB_YESNO + MB_ICONQUESTION, _
        "Uniform access is in effect") = IDYES Then
        acl.UniformAccess = False
        Call acl.Save
    End If
Else
    If MessageBox("Do you want to toggle?", _
        MB_YESNO + MB_ICONQUESTION, _
        "Uniform access is not in effect") = IDYES Then
        acl.UniformAccess = True
        Call acl.Save
    End If
End If
```

AddRole 成员函数

向 ACL 中增加一个角色。

定义

NotesACL

语法

Call notesACL.AddRole(name\$)

参数

name\$

字符串, 新角色的名字。在名称周围不使用圆括号或方括号。

样例: AddRole 成员函数

本 script 在当前数据库的 ACL 中增加一个 Developer 角色。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Set db = session.CurrentDatabase
Set acl = db.ACL
```

```
Call acl.AddRole( "Developer" )  
Call acl.Save
```

CreateACLEntry 成员函数

使用您所指定的名字和级别在 ACL 中创建一个项目。当使用 OLE 自动控制时，该成员函数允许您不使用 New 创建一个 NotesACLEntry 对象。

定义

NotesACL

语法

```
Set notesACLEntry = notesACL.CreateACLEntry( name$, level% )
```

参数

name\$

字符串，您要在 ACL 中为其创建项目的个人、群组或服务器的名字。

level%

常量，要在 ACL 中赋予指定的个人、群组或服务器的级别。有如下的可能性：

ACLLEVEL_NOACCESS	不能存取者
ACLLEVEL_DEPOSITOR	存放者
ACLLEVEL_READER	读者
ACLLEVEL_AUTHOR	作者
ACLLEVEL_EDITOR	编辑者
ACLLEVEL_DESIGNER	设计者
ACLLEVEL_MANAGER	管理者

返回值

notesACLEntry

新创建的项目。

用法

如果您希望把修改过的 ACL 保存到磁盘上，必须调用 ACL 中的 Save。

有关的详细信息，请参阅 NotesACLEntry 类。

样例: CreateACLEntry 成员函数

本 script 赋予 Shelly McPhail 数据库的编辑者权限。

```
Dim db As New NotesDatabase( "Warsaw", "data\discuss.nsf" )
Dim acl As NotesACL
Dim entry As NotesACLEntry
Set acl = db.ACL
Set entry = acl.CreateACLEntry _
( "Shelly McPhail", ACLLEVEL_EDITOR )
Call acl.Save
```

DeleteRole 成员函数

从 ACL 中删除一个角色。

定义

NotesACL

语法

```
Call notesACL.DeleteRole( name$ )
```

参数

name\$

字符串, 要删除的角色名字。

用法

如果您指定的角色在 ACL 中不存在, Notes 指出错误信息“没有找到角色名”。

样例: DeleteRole 成员函数

本样例从当前数据库的 ACL 中删除 Developer 角色。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Set db = session.CurrentDatabase
Set acl = db.ACL
Call acl.DeleteRole( "Developer" )
Call acl.Save
```

GetEntry 成员函数

给出一个名字, 在 ACL 中寻找它的项目。

定义

NotesACL

语法

```
Set notesACLEntry = notesACL.GetEntry( name$ )
```

参数

name\$

字符串, 您要查找的 ACL 项目名。该名字必须精确匹配, 区分大小写。

返回值

notesACLEntry

与 name\$ 匹配的 ACL 项目, 如果 name\$ 没有在 ACL 中, 则 Nothing。

用法

GetEntry 可以找出 ACL 中的个人、群组或服务器。如果个人没有列在 ACL 中, 但是它是列在 ACL 中的群组的成员, GetEntry 将不能找出该个人的名字。如果您需要这种功能, 请使用 NotesDatabase 中的 QueryAccess 成员函数。

样例: GetEntry 成员函数

1. 本 script 从 JUDGE.NSF 中的 ACL 中取得 Kendra Bauer 的项目。

```
Dim db As New NotesDatabase( "Kabul", "ubthe\judge.nsf" )
Dim acl As NotesACL
Dim entry As NotesACLEntry
Set acl = db.ACL
Set entry = acl.GetEntry( "Kendra Bauer" )
```

2. 本 script 从当前数据库取得 Destroyers 群组的项目。它修改 ACL 使 Destroyers 成员不能删除数据库中的文档。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Dim entry As NotesACLEntry
Set db = session.CurrentDatabase
Set acl = db.ACL
Set entry = acl.GetEntry( "Destroyers" )
If Not ( entry Is Nothing ) Then
    entry.CanDeleteDocuments = False
    Call acl.Save
End If
```

GetFirstEntry 成员函数

返回 ACL 中的第一个项目，通常是 -Default- 项目。

定义

NotesACL

语法

Set *notesACLEntry* = *notesACL*.GetFirstEntry

返回值

notesACLEntry

ACL 中的第一个项目。

样例: GetFirstEntry 成员函数

本 script 从一个叫做 HELLO.NSF 的数据库中获取 ACL 的第一个项目。

```
Dim db As New NotesDatabase( "Kampala", "hello.nsf" )
Dim acl As NotesACL
Dim firstEntry As NotesACLEntry
Set acl = db.ACL
Set firstEntry = acl.GetFirstEntry
```

GetNextEntry 成员函数

给出一个 ACL 项目，返回下一个。

定义

NotesACL

语法

Set *notesACLEntry* = *notesACL*.GetNextEntry(*notesACLEntry*)

参数

notesACLEntry

ACL 中的任意项目。

返回值

notesACLEntry

该项目跟在 ACL 中指定的 *notesACLEntry* 参数之后。如果没有其他的项目，则返回 Nothing。

样例: **GetNextEntry** 成员函数

本 script 从 HELLO.NSF 的 ACL 中获取第二个项目。

```
Dim db As New NotesDatabase( "Kampala", "hello.nsf" )
Dim acl As NotesACL
Dim firstEntry As NotesACLEntry
Dim secondEntry As NotesACLEntry
Set acl = db.ACL
Set firstEntry = acl.GetFirstEntry
Set secondEntry = acl.GetNextEntry( firstEntry )
```

RenameRole 成员函数

改变角色的名字。

定义

NotesACL

语法

Call **notesACL.RenameRole**(*oldName\$*, *newName\$*)

参数

oldName\$

字符串，角色的当前名称。

newName\$

字符串，您想给该角色的新名字。

用法

当您重新命名一个角色时，ACL 中拥有旧角色的所有成员都将获得新的角色。

样例: **RenameRole** 成员函数

本样例将 HR Contact 角色重新命名为 Human Resources 角色。任何以前拥有 HR Contact 角色的项目现在都变成 Human Resources 角色。

```
Dim session As New NotesSession
Dim db As NotesDatabase
Dim acl As NotesACL
Set db = session.CurrentDatabase
Set acl = db.ACL
```

```
Call acl.RenameRole( "HR Contact", "Human Resources" )  
Call acl.Save
```

Save 成员函数

保存对 ACL 所作的修改。如果在关闭数据库之前没有调用 Save，则对 ACL 所作的修改将丢失。

定义

NotesACL

语法

Call notesACL.Save

样例: Save 成员函数

1. 本 script 在当前数据库的存取控制表中添加一个新的角色，并且保存 ACL。如果 script 没有调用 Save，该角色将不被增加。

```
Dim session As New NotesSession  
Dim db As NotesDatabase  
Dim acl As NotesACL  
Set db = session.CurrentDatabase  
Set acl = db.ACL  
Call acl.AddRole( "HR Contact" )  
Call acl.Save
```

2. 本 script 提示用户输入成员的名称。本 script 在当前数据库中的 ACL 中为该成员创建新的项目，并且赋给他作者权限。然后该 script 询问用户来确认 ACL 的修改。如果用户选取“是”，则保存修改。如果用户选取“否”，则不保存修改。

要使用与 Messagebox 函数相关联的常量，该 script 必须包含文件 LSCONST.LSS。

```
Dim session As New NotesSession  
Dim db As NotesDatabase  
Dim acl As NotesACL  
Dim entry As NotesACLEntry  
Dim newName As String  
Dim confirm As Integer  
Set db = session.CurrentDatabase  
Set acl = db.ACL  
newName = Inputbox$( "Who do you want to add?", "Name" )  
Set entry = New NotesACLEntry( acl, newName, ACLLEVEL_AUTHOR )  
confirm = Messagebox( "Are you sure?", MB_YESNO, "Confirm" )  
If ( confirm = IDYES ) Then  
    Call acl.Save  
End If
```